Statistics 231B SAS Practice Lab #6
Spring 2006

This lab is designed to give the students practice in learning strategy for building a regression models whose steps are outlined in Figure 9.1 (page 344) of textbook ALSM.
Example:  A personnel officer in a governmental agency administered four newly developed aptitude tests to each of 25 applicants for entry-level clerical positions in the agency. For purpose of the study, all 25 applicants were accepted for positions irrespective of their test scores. After a probationary period, each applicants was rated for proficiency on the job. The scores on the four tests (X1,X2,X3,X4) and the job proficiency score (Y) for the 25 employees were recorded in the file CH09PR10.txt. 
1. Prepare separate box plots of the test scores for each of the four newly developed aptitude tests. Are there any noteworthy features in these plots (eg. symmetry of the distribution, possible outliers)?  Comment.

SAS CODE:

data jobprof;

infile "c:\stat231B06\ch09pr10.txt";

input y x1 x2 x3 x4 ;
x1x3=x1*x3;

x1x4=x1*x4;

x3x4=x3*x4;
run;

/*prepare separate box plots of the test scores for X1,X2,X3 and X4*/
/*PROC UNIVARIATE can be used to get box plot as we showed in lab 2*/
/*alternatively, you can also use PROC BOXPLOT to get box plot*/
/*PROC BOXPLOT produces high quality graphics which can be copied and inserted in a report or research paper.*/
/*PROC BOXPLOT can be used to produce side by side boxplots of a variable broken down by one or more categories. */
/*For example, we might wish to compare the distribution of income by ethnic groups.*/ 

/*for details, refer to following SAS/STAT9.1 user guide Chapter 18.*/

/*http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc_91/stat_ug_7313.pdf*/

/*However, PROC BOXPLOT does not do one variable boxplot itself. */
/*So in this example, we must create a variable which is any constant value and here we call it as cvar*/
/*Then we use PROC BOXPLOT to plot X1 by cvar, X2 by cvar, X3 by cvar and X4 by cvar.*/
/*the outliers will be shown if boxstyle=schematic*/
data jobprof;

set jobprof;

cvar=0;/*our second variable to plot by;*/
run;

proc boxplot;

/*changing the value of boxwidth will change the width of the box in the box plot*/
plot (x1 x2 x3 x4)*cvar / boxstyle=schematic boxwidth=10;  

run;

quit;
SAS output:
X1                                                                     X2

X3                                                                 X4
The distribution of X1 and X2 are a little bit skewed. There is a potential outlier in X1.
2. Obtain the scatter plot matrix. Also obtain the correlation matrix of the X variables. What do the scatter plots suggest about the nature of the functional relationship between the response variable Y and each of the predictor variable? Are any serious multicollinearity problem evident? Explain.
SAS CODE:

  proc insight data=jobprof;

   scatter Y x1 x2 x3 x4 * Y x1 x2 x3 x4;

   run;

   proc corr;

   var Y x1 x2 x3 x4;

   run;
SAS OUTPUT:

     The CORR Procedure

                  5  Variables:    y        x1       x2       x3       x4

                                       Simple Statistics

   Variable           N          Mean       Std Dev           Sum       Minimum       Maximum

   y                 25      92.20000      19.42292          2305      58.00000     127.00000

   x1                25     103.36000      20.29754          2584      62.00000     150.00000

   x2                25     106.72000      17.29287          2668      73.00000     129.00000

   x3                25     100.80000       8.85532          2520      80.00000     116.00000

   x4                25      94.68000      10.67599          2367      74.00000     110.00000

                           Pearson Correlation Coefficients, N = 25

                                   Prob > |r| under H0: Rho=0

                           y            x1            x2            x3            x4

            y        1.00000       0.51441       0.49701       0.89706       0.86939
                                    0.0085        0.0115        <.0001        <.0001

            x1       0.51441       1.00000       0.10227       0.18077       0.32666

                      0.0085                      0.6267        0.3872        0.1110

            x2       0.49701       0.10227       1.00000       0.51904       0.39671

                      0.0115        0.6267                      0.0078        0.0496

            x3       0.89706       0.18077       0.51904       1.00000       0.78204
                      <.0001        0.3872        0.0078                      <.0001

            x4       0.86939       0.32666       0.39671       0.78204       1.00000

                      <.0001        0.1110        0.0496        <.0001
Based on the correlation values highlighted in green and the first row of the scatter plot matrix, Y is highly linearly correlated with all four predictor variables. 
Based on the correlation values highlighted in blue and the corresponding scatter plots in the scatter plot matrix, X2 and X3, X3 and X4 are highly correlated which suggests the possibility of multicollinearity problem.
(3) Fit the multiple regression function containing all four predictor variables as first-order terms. Does it appear that all predictor variables should be retained?
SAS CODE:

proc reg data=jobprof;

model y=x1 x2 x3 x4;

run;

SAS OUTPUT:

     The REG Procedure

                                         Model: MODEL1

                                     Dependent Variable: y

                            Number of Observations Read          25

                            Number of Observations Used          25

                                      Analysis of Variance

                                             Sum of           Mean

         Source                   DF        Squares         Square    F Value    Pr > F

         Model                     4     8718.02248     2179.50562     129.74    <.0001

         Error                    20      335.97752       16.79888

         Corrected Total          24     9054.00000

                      Root MSE              4.09864    R-Square     0.9629

                      Dependent Mean       92.20000    Adj R-Sq     0.9555

                      Coeff Var             4.44538

                                      Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1     -124.38182        9.94106     -12.51      <.0001

              x1            1        0.29573        0.04397       6.73      <.0001

              x2            1        0.04829        0.05662       0.85      0.4038
              x3            1        1.30601        0.16409       7.96      <.0001

              x4            1        0.51982        0.13194       3.94      0.0008

It seems X2 should be excluded from the model since the p-value=0.4038.
(4) Obtain a plot of residuals against predicted vales and normal probability plot for the fitted model obtained in (3). Do you see any problems?

SAS CODE:

proc reg data=jobprof;

model y=x1 x2 x3 x4;

/*r.*p. means residual versus predicted value*/

/* r.*npp. means normal probability plot*/
plot r.*p. r.*npp.; 

run;
SAS OUTPUT:

The plots suggests the constant error variance and normality of residual. Therefore, no transformation is needed.
(5) Using only first-order terms for the predictor variables in the pool of potential 4 x variables, find the four best subset regression models according to the 

SAS CODE:
  proc reg data=jobprof;

   model y=x1 x2 x3 x4/selection = adjrsq;

  run;
SAS OUTPUT

       The REG Procedure

                                         Model: MODEL1

                                     Dependent Variable: y

                               Adjusted R-Square Selection Method

                            Number of Observations Read          25

                            Number of Observations Used          25

                    Number in    Adjusted

                      Model      R-Square    R-Square    Variables in Model

                           3       0.9560      0.9615    x1 x3 x4

                           4       0.9555      0.9629    x1 x2 x3 x4

                           2       0.9269      0.9330    x1 x3

                           3       0.9247      0.9341    x1 x2 x3
                           2       0.8661      0.8773    x3 x4

                           3       0.8617      0.8790    x2 x3 x4

                           3       0.8233      0.8454    x1 x2 x4

                           2       0.7985      0.8153    x1 x4

                           1       0.7962      0.8047    x3

                           2       0.7884      0.8061    x2 x3

                           2       0.7636      0.7833    x2 x4

                           1       0.7452      0.7558    x4

                           2       0.4155      0.4642    x1 x2

                           1       0.2326      0.2646    x1

                           1       0.2143      0.2470    x2

The four best subset regression models according to the 

(X1,X3,X4) (X1,X2,X3,X4) (X1 X3) (X1 X2 X3). 

(6) Since there is relatively little difference in 

 for the four best subset models, what other criteria would you use to help in the selection of the best model?
SAS CODE:

proc reg data=jobprof;

   model y=x1 x2 x3 x4/selection = adjrsq cp aic sbc;

run;

A few notes:

(1) depending on which option is specified first on the right hand side of the equation “selection=”, SAS will use that option as the selection method and will report values for all other options specified after the first option. For example, selection=adjrsq cp aic sbc means selecting best subset model using adjusted R square as criterion, report values in order for cp, aic and sbc.

(2) In SAS 9.1 version that we are currently using, SAS doesn’t use AIC and SBC for selecting best subset model. Therefore, we can not  put “aic” or “sbc” as first option for “selection=”.  If we want to use AIC or SBC as selection method, we can specify slection=adjrsq aic sbc to get AIC and SBC value and then conduct selection by ourselves.
       Model: MODEL1

                                     Dependent Variable: y

                               Adjusted R-Square Selection Method

                            Number of Observations Read          25

                            Number of Observations Used          25

Number in    Adjusted

  Model      R-Square    R-Square        C(p)           AIC            SBC    Variables in Model

       3       0.9560      0.9615      3.7274       73.8473       78.72282    x1 x3 x4
       4       0.9555      0.9629      5.0000       74.9542       81.04859    x1 x2 x3 x4

       2       0.9269      0.9330     17.1130       85.7272       89.38384    x1 x3

       3       0.9247      0.9341     18.5215       87.3143       92.18984    x1 x2 x3

       2       0.8661      0.8773     47.1540      100.8605      104.51716    x3 x4

       3       0.8617      0.8790     48.2310      102.5093      107.38479    x2 x3 x4

       3       0.8233      0.8454     66.3465      108.6361      113.51157    x1 x2 x4

       2       0.7985      0.8153     80.5653      111.0812      114.73788    x1 x4

       1       0.7962      0.8047     84.2465      110.4685      112.90629    x3

       2       0.7884      0.8061     85.5196      112.2953      115.95191    x2 x3

       2       0.7636      0.7833     97.7978      115.0720      118.72864    x2 x4

       1       0.7452      0.7558    110.5974      116.0546      118.49234    x4

       2       0.4155      0.4642    269.7800      137.7025      141.35916    x1 x2

       1       0.2326      0.2646    375.3447      143.6180      146.05576    x1

       1       0.2143      0.2470    384.8325      144.2094      146.64717    x2

The C(p), AIC and SBC criteria all suggest that model with (X1,X3,X4) as predictor variables is the best subset model
(7) using forward stepwise regression, find the best subset of predictor variables to predict job proficiency. Use (=0.05 and 0.10 for adding or deleting a variable, respectively. 

SAS CODE:

proc reg data=jobprof;

/*sle specifies the significance level for adding a variable, sls specifies the significance level for deleting a variable*/
model y=x1 x2 x3 x4/selection=stepwise sle=0.05 sls=0.1;

run;
                                 Summary of Stepwise Selection
          Variable    Variable    Number    Partial     Model

   Step   Entered     Removed     Vars In   R-Square   R-Square    C(p)     F Value   Pr > F

     1    x3                          1      0.8047     0.8047    84.2465     94.78   <.0001

     2    x1                          2      0.1283     0.9330    17.1130     42.12   <.0001

     3    x4                          3      0.0285     0.9615     3.7274     15.59   0.0007
The forward stepwise model selection method suggests that model with (X1,X3,X4) as predictor variables is the best subset model which is consistent with the conclusion we draw earlier 
(8) Based on the results obtained in (6) and (7), we want to evaluate the subset model containing only first-order terms in X1,X3 and X4 in detail. Obtain the residuals and plot them separately against 

, each of the predictor variables in the model, and each of the related cross-product terms. On the basis of these plots, should any modifications of the model be investigated?
SAS CODE:
proc reg data=jobprof;

   model y=x1 x3 x4;

/*the new dataset "result" contains all the columns in data set "jobprof" and two new columns "resid" and "pred"*/
   output out=result r=resid p=pred;

   run;

goptions reset=all; /*reset all the settings for graphics to default*/ 

proc gplot data=result;
/*proc gplot same as proc plot but with more potions and nicer graph*/

plot resid*(pred x1 x3 x4 x1x3 x1x4 x3x4);

run;

SAS OUTPUT:

Residual vs. fitted value

Doesn’t see to have curvilinear effect or interaction. 

(9) Prepare separate added-variable plots of e(Y|X3) vs e(X4|X3), 

e(Y|X1) vs e(X3|X1), e(Y|X3) vs e(X1|X3). Do these plots suggest that any modifications in the model form are warranted?

SAS CODE:

proc reg data=jobprof;

model y=x3;

output out=result1 r=residyx3; /*obtain e(Y|X3)*/
model x4=x3;

output out=result2 r=residx4x3;/*obtain e(X4|X3)*/
run;

proc sort data=result1; by y;run; /*sort result1 by Y*/
proc sort data=result2; by y;run; /*sort result2 by Y*/
data resulta; /*merge the two datasets together containing*/
                   /*one column of residyx3 and one column of residx4x3*/
merge result1 result2;

by y;

run;

goptions reset=all; /*reset all the settings for graphics to default*/ 

symbol1 value=circle color=black interpol=rl;

proc gplot data=resulta;

plot residyx3*residx4x3;

run;
SAS OUTPUT:

the plots of e(Y|X1) vs e(X3|X1) and  e(Y|X3) vs e(X1|X3) can be obtained similarly. (code not shown, but output shown below)

e(Y|X1) vs e(X3|X1)

e(Y|X3) vs e(X1|X3)

The marginal contribution of each predictor seems linear and no need for modification
(10) Prepare a normal probability plot of the residuals. Test the reasonableness of the normality assumptions using Shapiro-Wilk test (Shapiro-Wilk test is in the reference 3.3 on page 146 of textbook ALSM).
SAS CODE:

proc reg data=jobprof;

model y=x1 x3 x4;

plot r.*npp.;/*obtain normal probability plot*/
output out=result r=resid;

run;

/*obtain Shapiro-Wilk normality test*/
proc univariate data=result normal;

var resid; 

run;

SAS OUTPUT

The UNIVARIATE Procedure

                                  Variable:  resid  (Residual)

                                            Moments

                N                          25    Sum Weights                 25

                Mean                        0    Sum Observations             0

                Std Deviation      3.80896425    Variance            14.5082087

                Skewness            0.1642578    Kurtosis            -1.0469646

                Uncorrected SS     348.197008    Corrected SS        348.197008

                Coeff Variation             .    Std Error Mean      0.76179285

                                   Basic Statistical Measures

                         Location                    Variability

                     Mean      0.00000     Std Deviation            3.80896

                     Median   -0.20567     Variance                14.50821

                     Mode       .          Range                   12.06619

                                           Interquartile Range      4.96332

                                   Tests for Location: Mu0=0

                        Test           -Statistic-    -----p Value------

                        Student's t    t         0    Pr > |t|    1.0000

                        Sign           M      -0.5    Pr >= |M|   1.0000

                        Signed Rank    S       0.5    Pr >= |S|   0.9896

                                      Tests for Normality

                   Test                  --Statistic---    -----p Value------

                   Shapiro-Wilk          W     0.941495    Pr < W      0.1603
                   Kolmogorov-Smirnov    D     0.107068    Pr > D     >0.1500

                   Cramer-von Mises      W-Sq  0.046714    Pr > W-Sq  >0.2500

                   Anderson-Darling      A-Sq  0.397585    Pr > A-Sq  >0.2500

Based on Normal Probability plot and formal Sahpiro-Wilk test (pvalue=0.1603), the normal distribution assumption is valid assumption.
(11) 
(a) obtain the studentized deleted residuals and identify any outlying Y observations. Use the Bonferroni outlier test procedure with (=0.05. State the decision rule and conclusion

(b) obtain the diagonal elements of the hat matrix. Using the rule of thumb in the textbook, identify any outlying X observations. 

(c) using DFFITS, DEBETAS and Cook’s distance to assess the influence the outlying observations you identified in (a) and (b). What do you conclude?
SAS CODE:

  proc reg data=jobprof;

/*obtain studentized deleted residuals and hat matrix*/
   model y=x1 x3 x4/influence; 

  /*output Cook distance, DFFITS*/
   output out=result1 cookd=cookd dffits=dffits;

   ods output outputstatistics=result2;

   run;

/*print out cook’s distance values*/

proc print data=result1;

var cookd;

run;

/*obtain t critical value for Bonferroni outlier test*/  

data t;

   tvalue=tinv(0.999,20);/0.999=1-0.05/(2*25),20=25-4-1*/

   run;

   proc print data=t;

   run;
/*obtain F percentile based on Cook’s distance*/

data result1;

   set result1;

   percent1=100*probf(cookd,4,21);

   run;

proc print data=result1;

   var percent1;

run
                                         Model: MODEL1

                                     Dependent Variable: y

                                      Output Statistics

                               Hat Diag      Cov          ---------------DFBETAS--------------

       Obs  Residual  RStudent        H    Ratio   DFFITS Intercept       x1       x3       x4

         1    6.0036    1.6110   0.0988   0.8279   0.5333    0.1099  -0.1956   0.2355  -0.2846

         2   -0.2541   -0.0751   0.3420   1.8452  -0.0541   -0.0181   0.0466   0.0237  -0.0331
         3   -5.4579   -1.4313   0.0793   0.8938  -0.4200    0.0469  -0.0010   0.1748  -0.2721

         4   -5.0819   -1.3670   0.1320   0.9797  -0.5331   -0.3048   0.0958   0.4432  -0.3583

         5    0.1224    0.0303   0.0593   1.2918   0.0076    0.0045   0.0000  -0.0017  -0.0011

         6    1.7113    0.4397   0.1217   1.3316   0.1637    0.0926  -0.0918   0.0078  -0.0438

         7   -0.2057   -0.0612   0.3501   1.8689  -0.0449   -0.0302  -0.0213   0.0178   0.0088
         8   -1.0994   -0.2970   0.2094   1.5105  -0.1528    0.1086  -0.0061  -0.1288   0.0703

         9   -3.1563   -0.8022   0.0823   1.1669  -0.2402    0.0856  -0.0141   0.0441  -0.1284

        10   -5.0305   -1.3119   0.0828   0.9525  -0.3942    0.1708  -0.2408  -0.1660   0.1385

        11   -5.2405   -1.4207   0.1396   0.9617  -0.5723   -0.4592   0.2475   0.3948  -0.2317

        12    1.5698    0.4180   0.1827   1.4364   0.1976   -0.1480   0.1138   0.0748  -0.0098

        13    0.3950    0.1082   0.2346   1.5844   0.0599    0.0007   0.0527   0.0016  -0.0201

        14   -3.0330   -0.8448   0.2331   1.3776  -0.4658   -0.0892   0.0566  -0.2767   0.3790

        15   -0.5189   -0.1328   0.1218   1.3792  -0.0494    0.0204   0.0102   0.0070  -0.0304

        16    6.6083    1.9067   0.1846   0.7643   0.9072    0.0191  -0.1339  -0.5934   0.7931

        17    4.9084    1.2956   0.1063   0.9854   0.4469    0.0739  -0.0934  -0.2908   0.3518

        18    4.6393    1.3808   0.2897   1.1890   0.8819   -0.3660   0.6235  -0.1456   0.2655

        19   -1.5474   -0.3941   0.1076   1.3206  -0.1369    0.0568   0.0111  -0.1056   0.0756

        20   -4.8389   -1.2686   0.0971   0.9878  -0.4161   -0.1213  -0.2625   0.1021   0.0486

        21    1.8070    0.4759   0.1626   1.3875   0.2097    0.1592  -0.1389  -0.0848   0.0353

        22   -3.2360   -0.8607   0.1581   1.2483  -0.3730    0.2335   0.0914  -0.2570   0.0855

        23    3.7904    1.0013   0.1356   1.1563   0.3965    0.1689   0.1215   0.0835  -0.2674

        24    1.4823    0.3934   0.1781   1.4340   0.1831   -0.1144  -0.0538   0.1230  -0.0354

        25    5.6625    1.5200   0.1107   0.8827   0.5364    0.1764  -0.0595   0.2158  -0.3637

                          Sum of Residuals                           0

                          Sum of Squared Residuals           348.19701

                          Predicted Residual SS (PRESS)      471.45198
                                          Obs     cookd

                                           1    0.06609

                                           2    0.00077
                                           3    0.04201

                                           4    0.06823

                                           5    0.00002

                                           6    0.00696

                                           7    0.00053
                                           8    0.00610

                                           9    0.01467

                                          10    0.03756

                                          11    0.07810

                                          12    0.01017

                                          13    0.00094

                                          14    0.05499

                                          15    0.00064

                                          16    0.18283

                                          17    0.04837

                                          18    0.18639

                                          19    0.00488

                                          20    0.04206

                                          21    0.01141

                                          22    0.03522

                                          23    0.03930

                                          24    0.00873

                                          25    0.0677
Obs     tvalue

                                          1     3.23149
percentile of F distribution based on Cook’s distance

Obs    percent1

                                          1     0.86250

                                          2     0.00013
                                          3     0.36180

                                          4     0.91634

                                          5     0.00000

                                          6     0.01051

                                          7     0.00006
                                          8     0.00808

                                          9     0.04608

                                         10     0.29120

                                         11     1.18236

                                         12     0.02227

                                         13     0.00019

                                         14     0.60743

                                         15     0.00009

                                         16     5.52646

                                         17     0.47492

                                         18     5.71322

                                         19     0.00518

                                         20     0.36261

                                         21     0.02802

                                         22     0.25700

                                         23     0.31800

                                         24     0.01648

                                         25     0.90294

(a) case 16 appears to be moderately outlying with respect to their Y values. However the value 1.9067 is less than the Bonferroni critical value 3.23149. We conclude case 16 is not an outlier.
(b) since 2p/n=2*4/25=0.32 and leverage values greater than 2p/n are considered to indicate outlying cases with regard to their X values, case 2 (H value=0.3420) and case 7 (H value=0.3501) are concluded as outlying X observations.
       (c)Case 2 with Cook’s distance value=0.0007 is the 0.013 percentile of the F-
           distribution and case 7 with Cook’s distance value=0.00053 is the 0.006 
            percentile of the F-distribution. 
2*sqrt(p/n)=0.8 and absolute values of DFFITS values for case 2 and   

7 are 0.0541 and 0.0449, both are less than 0.8.
DFBETAS values for case 2 and 7 are all much less than 1.

Based on all three criteria, we conclude that all the outlying X observations are not influential.
(12) Obtain the variance inflaction factors. What do they indicate?

SAS CODE:

proc reg data=jobprof;

   model y=x1 x3 x4/VIF; 

   run;
SAS OUTPUT:

      The REG Procedure

                                         Model: MODEL1

                                     Dependent Variable: y

                            Number of Observations Read          25

                            Number of Observations Used          25

                                      Analysis of Variance

                                             Sum of           Mean

         Source                   DF        Squares         Square    F Value    Pr > F

         Model                     3     8705.80299     2901.93433     175.02    <.0001

         Error                    21      348.19701       16.58081

         Corrected Total          24     9054.00000

                      Root MSE              4.07195    R-Square     0.9615

                      Dependent Mean       92.20000    Adj R-Sq     0.9560

                      Coeff Var             4.41644

                                      Parameter Estimates

                           Parameter       Standard                              Variance

      Variable     DF       Estimate          Error    t Value    Pr > |t|      Inflation

      Intercept     1     -124.20002        9.87406     -12.58      <.0001              0

      x1            1        0.29633        0.04368       6.78      <.0001        1.13775
      x3            1        1.35697        0.15183       8.94      <.0001        2.61664
        x4            1        0.51742        0.13105       3.95      0.0007        2.83349
Since all three VIF don’t exceed 10, which indicates no serious multicollinearity problems exist.

